Contract No. E(34-1)-0003
Cities Service Oil Company

report number "BERC-0003-8"

EL DORADO MICELLAR - POLYMER PROJECT

TECHNICAL LETTER FOR MARCH, 1976

MASTER

Summary of Activities for March, 1976

The well stimulation program that was initiated last month has continued to appear quite favorable. The program was extended to include all injection wells operating below 100 barrels per day.

A total of 20,280 and 19,160 barrels of pretreatment fluids were injected into the Chesney and Hegberg pattern, respectively.

Work concerned with the quality of injection water has continued on site and in the laboratory. Additional information has been received about the formulation of the Shell system. Laboratory flow experiments, modeling and other work directed toward questions raised by study of this information has been initiated.

Meetings were held with representatives of Union and Witco to discuss the blending and specifications for petroleum sulfonates used in the Union process. Mobility was the topic of a joint meeting with Union and Shell. As a result of the joint meeting, additional relative permeability work has been undertaken. Many aspects of the Union design were reviewed in a Cities Service-Union Oil meeting on March 24.

The streamline and front tracking model has been documented using comment statements inserted within the program listing. A copy of the program is enclosed. This model has been used to compare sweep for various patterns and operating conditions. Polymer slug simulation has continued.

Technical papers. Three technical papers relating to the El Dorado project were presented at the Society of Petroleum Engineers of AIME Fourth Symposium on Improved Oil Recovery on March 23. The papers report core analysis, performance prediction, and pressure transient work.

DISCLAIMER

ENGINEERING AND OPERATIONS

Pattern Injectivity and Well Stimulation

The Xylene-Hydrochloric Acid- Hydroflouric Acid- Hydrochloric Acid treatment that was initiated last month has continued to appear favorable. As a result of this treatment, it was decided to acidize all the injection wells operating below 100 barrels per day. Below is a summary of the treatment performed this month:

WELL	BEFORE B/D	PRESSURE(PSI)	AFTER	PRESSURE(PSI)	RATE NOW B/D	PRESSURE(PSI)
MP108	27	180	151	150	102	185
MP110	63	195	165	60	158	80
MP118	12	185	138	185	52	185
MP120	78	190	146	170	135	190
MP128	97	190	137	130	130	190
MP130	2	190	154	115	92	190
MP213-226	33	210	85	200	77	200
MP215	32	200	151	80	126	195
MP221	62	200	151	55	150	190
MP223	40	200	146	185	91	195
MP225	35	215	150	195	97	210

Work is continuing to stimulate additional wells.

In order to control the bacteria growing in the supply water, a chlorine gas was introduced at the pump station. The Fresh Water Line and Plant Facilities was initially slugged with a high chlorine concentration (20-30PPM) Before adjusting to 7 PPM at the pump station. Future plans will be to maintain a chlorine residual between 0.15 - 0.25.

MONTHLY PRODUCTION REPORT

Well No.	Monthly Volumes, Bbls. Oil	Water	Days Prod.	Cumul. Since Start Oil	Water	Date Started
MP 112	37	1468	31	105	4958	11/17/75
MP 114	76	785	31	219	1999	"
MP 122	50	1113	31	109	3422	"
MP 124	64	1984	31	156	6459	"
Chesney	227	5350		589	16838	
MP 207	37	814	31	271	2390	"
MP 209	51	1326	31	102	4232	"
MP 217	38	1621	31	97	5047	"
MP 219	38	1469	31	103	4952	"
Hegberg	164	5230		573	16621	

MONTHLY INJECTION REPORT (Pretreatment)

Well No.	Barrels Injected	Injection Press. psig	Days on	Cumulative Inj. bbls.	Date Started
MP 106	1948	190	31	4395	11/17/75
MP 108	2693	190	30	5773	"
MP 110	1665	190	20	7847	"
MP 116	3544	190	31	6055	"
MP 118	1265	190	31	4765	"
MP 120	2744	190	30	9507	"
MP 126	1513	190	25	7837	"
MP 128	2761	190	28	6919	"
MP 130	2147	190	30	4043	"
Chesney	20,280			57,141	
MP 201	1,571	200	26	8,494	"
MP 203	3,082	200	31	6,477	"
MP 205	3,314	200	31	6,058	"
MP 211	1,383	200	25	7,434	"
MP 213-226	530	200	17	5,330	"
MP 215	2,575	200	28	5,713	"
MP 221	2,927	200	29	7,366	"
MP 223	1,549	200	25	5,892	"
MP 225	2,229	200	29	6,156	"
Hegberg	19,160			58,920	
Total Project	39,440			116,061	

MONTHLY PRODUCING WELL TEST REPORT

Well No.	Actual Vol. Produced Oil	Water	Hours Tested	Pump Size	Stroke Length	Strokes Per Min.
MP 112	Tr.	160	24	1½	42"	17.
MP 114	5.	65	24	1½	32"	13.5
MP 122	5.	160	24	1½	42"	16.
MP 124	0.	250	24	1½	42"	14.5
MP 207	10.	80	24	1½	22"	13.5
MP 209	Tr.	145	24	1½	42"	16.5
MP 217	Tr.	95	24	1½	42"	16.
MP 219	5.	150	24	1½	42"	15.

RESEARCH SUPPORT

Chemical Selection and Support

Water Quality. A considerable amount of time was devoted to water quality during March. In addition to the laboratory work, one man was on site for a week. The prime objective of the field work was to see if the water quality could be improved by using sodium bentonite to flocculate the particles in the water so that they are more readily filtered out.

The experiment with bentonite had been tried in the Laboratory, but the particle distribution had changed in the El Dorado water at the Laboratory. It was found that most of the particles had increased to above 1.2 μ in size. This increase in particle size is additional evidence that bacteria has been a cause of the water quality problem.

The field data are summarized in the table below. The water quality standards for these tests were given in the monthly report for February in the section on "Chemical Selection and Support." Briefly, good quality water should have an initial flow rate of about 10 ml/sec with a rate still above 4 ml/sec after three liters have passed through a 0.45 μ filter. The last column in the table lists volumes filtered; in some cases the filter plugged so much that less than three liters were filtered. The most obvious conclusion is that treating the water with bentonite very definitely improved the quality. Some comments on the various experiments follow.

Experiments 1, 2, and 3 and Experiments 6, 7, and 8 were made to determine particle size. It seems that most particles not passing through the o.45 μ filter were less than 1.2 μ. Chlorination seems to be beneficial. Comparison of Experiments 2 with 7 and 3 with 8, shows that chlorination helped. Experiment 9 confirms that the water treated with bentonite is of excellent quality. As a consequence of these experiments, it was recommended to the Project Engineer that either different polymers be tried to find one which will work or that the water be treated with sodium bentonite.

WATER QUALITY

All Water Samples Taken Just Downstream From the Sand Filter

Before Chlorination March 15-17, 1976

	Treatment to Water	Filter	Range of Flow Rate (ml/sec) Initial 200 ml	Final 200 ml	Total Volume Filtered (Liters)
1.	None	0.45 μ	7.3	0.91	0.8
2.	None	0.8 μ	28.6	1.7	2.4
3.	None	1.2 μ	40.0	18.5	3.0
4.	Filtered through 1.2 μ	0.45 μ	10.0	0.88	1.6
5.	Bentonite, then 1.2 μ filter	0.45 μ	9.8	3.3	3.0
	After Chlorination March 18-19, 1976				
6.	None	0.45 μ	11.2	0.98	2.2
7.	None	0.8 μ	37.0	14.7	3.0
8.	None	1.2 μ	48.0	33.0	3.0
9.	Bentonite, then 1.2 μ filter	0.45 μ	12.0	9.5	3.0

Shell process formulation. A supplementary report received from Shell during March was reviewed and discussed (see also the section "Performance Prediction"). Oil recovery flow tests in native El Dorado cores are now being conducted to investigate the recovery as a function of surfactant slug size (Shell formulation). One test was conducted with continuous chemical slug injection. Two more tests were conducted using 50 percent and 10 percent PV (pore volume) slug sizes. Another test is under way using a slug size of 25 percent PV. Preliminary results indicated that the chemical slug does move the residual oil and recovery increases with increasing slug size. Final results and evaluation should be available next month.

Mobility design. A meeting to discuss mobility design was held with representatives of Union and Shell. At that meeting it was decided to undertake additional work on the determination of total relative mobility (see also the section "Coring and Core analyses").

El Dorado model for IPE. Laboratory personnel have been preparing a physical model of the El Dorado Demonstration Flood so that it can be displayed at the International Petroleum Exposition in Tulsa this May. The model is made of plexiglass and packed with sand. The model is approximately six feet long by four feet wide by one-half inch thick. It is made approximately to scale representing the part of the El Dorado Field that is included in the demonstration test.

Other chemical work. A meeting was held with Union and Witco personnel on March 22. Quality control and performance testing of Witco's sulfonates were discussed.

Technical representatives from Union Oil Company and Cities Service met to review the project status and the Union design. Topics discussed included pore volume determination, oil recovery efficiency, chemical analysis, and polymer concentration.

Work in March also included injection water analyses and an analysis of water and solids swabbed from well MP-116. This work was part of the evaluation of injection well stimulation treatments.

Coring and Core Analyses

A paper entitled "Core Analysis Study for the El Dorado Micellar-Polymer Project" by W. H. Pusch was presented at the Society of Petroleum Engineers of AIME Improved Oil Recovery Symposium in Tulsa.

At a meeting to discuss fluid mobilities (as discussed earlier in the section "Chemical Selection and Support"), it was decided that additional data should be obtained for high permeability core samples. Steady-state relative permeability tests are now being run on native-state plugs from well MP-124. El Dorado produced water and oil are being used in these tests.

Pressure Transient Tests

A paper "Interference Testing for Reservoir Definition--The State of the Art" by S. C. Swift and L. P. Brown was presented at the SPE Improved Oil Recovery Symposium in Tulsa.

Performance Prediction

A paper entitled "Reservoir Simulation for the El Dorado Micellar-Polymer Project" by D. F. Zetik was given at the SPE Improved Oil Recovery Symposium in Tulsa.

Several days were spent in meetings with Cities Service, Shell, and Union employees (see also the section "Chemical Selection and Support"). The topics of primary concern to those working in performance prediction were pore volume determination, chemical slug size determination, and polymer slug concentration taper calculations.

Pattern sweep. The streamline and front tracking model has been used to compare sweep for various patterns and operating conditions against the sweep within one five-spot in an infinite array of repeating five-spots. An isolated five-spot with total injection rate equal to four times the production rate has a sweep that closely resembles the sweep in the infinite array. However, an isolated five-spot with total injection rate equal to total production rate has sweep quite different from the sweep in the infinite array. The sweep for the El Dorado Project with a two to one ratio of production to injection rates also closely resembles the sweep in the infinite array. The sweep for the project with our proposed well rates (overinjection operating conditions)is much closer to the sweep in the infinite array than to the "erroneous" sweep exhibited by an isolated five spot using the one to one ratio. A two to one ratio of production to injection rates gives a slightly better approximation to the sweep in the infinite array than the proposed rates. However, this slight improvement does not justify the increased project life and chemical requirements over the proposed operating conditions.

The difference between the sweep in the project and sweep in an infinite array of five-spots may make the recovery from the project different from the recovery for a field-wide expansion. However, these recovery differences are expected to be small and easily corrected for by mathematical simulation. Use of the stream tube recovery model is planned to study the differences in oil production schedule between the El Dorado Project and an infinite array of five-spots.

Polymer slug simulation. A number of one-dimensional runs were made with an Intercomp polymerflood simulator and with a second simulator which accurately describes the displacement of a polymer slug with water.

The runs are necessary in order to determine what value to use with the Intercomp simulator for the parameter which controls numerical dispersion. Other questions concerning the way in which polymer concentration and mobility are calculated with the Intercomp simulator also need to be answered before a final evaluation is complete.

Program documented. The streamline and front tracking model was documented by inserting comment statements in the program listing. These statements explain the data entry and the purpose of various sections of code. A copy of this program is enclosed.

CHESNEY PATTERN





HEGBERG
PATTERN
PRESSURE
RATE
CUM
CUM
0
5000
10000
15000
20000
25000
30000
35000
40000
45000
50000
RATE (STB/D)
0
200
400
600
800
1000
1200
1400
1600
1800
2000
PRESSURE
0
20
40
60
80
100
120
140
160
180
200
11.16
11.23
11.30
12.07
12.14
12.21
12.28
1.04
1.11
1.18
1.25
2.01
2.08
2.15
2.22
2.29
3.07
CALENDAR DATE

Miller


```
C    PROGRAM SFPLOT - STREAMLINE AND FRONT PLOTS.
C
C    THIS PROGRAM PLOTS FLOW STREAMLINES AND INJECTION SLUG FRONTS
C    FOR AN INFINITE, HOMOGENEOUS, CONSTANT THICKNESS RESERVOIR
C    WITH ISOTROPIC PERMEABILITY.  THE SLUG FRONT CALCULATION ASSUMES
C    PISTON LIKE DISPLACEMENT.  IMAGE WELLS MAY BE USED TO SIMULATE
C    BOUNDED RESERVOIRS.
C
C    WRITTEN BY D. F. ZETIK, CITIES SERVICE OIL COMPANY, 1975-76.
C
C    ALTHOUGH THIS PROGRAM HAS BEEN TESTED BY CITIES SERVICE, A THROUGH
C    REVIEW HAS NOT BEEN ATTEMPTED, AND THEREFORE THE PROGRAM IS MADE
C    AVAILABLE SUBJECT TO THE UNDERSTANDING THAT CITIES SERVICE DOES
C    NOT MAKE ANY WARRANTY, EXPRESSED OR IMPLIED, RESPECTING THIS
C    PROGRAM, INCLUDING BUT NOT BY WAY OF LIMITATION, THE IMPLIED
C    WARRANTY OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE,
C    AND SUBJECT TO THE UNDERSTANDING THAT CITIES SERVICE SHALL IN NO
C    WAY BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES.
C
      COMMON/PSCALE/XPPARM(4),YPPARM(4),XMIN,XMAX,YMIN,YMAX,SCALE,
     2 WPLOT,FPLOT
      COMMON/WDATA/Q(99),XW(99),YW(99),BTTIME(99),NWELLS,CONST,RI,RISQ
      DIMENSION ATOTAL(20),RBTOT(20),BITOT(20),BTTBAR(99)
      DIMENSION FPTIME(20),ITITLE(40)
      DIMENSION XFOLD(121),YFOLD(121),DXDTO(121),DYDTO(121),KPOLD(121)
      DIMENSION XFNEW(121),YFNEW(121),DXDTN(121),DYDTN(121),KPNEW(121),
     2 ANGLE(121)
      DIMENSION XCURVE(504),YCURVE(504)
      DOUBLE PRECISION DSUM1
      DATA PARM1/0.05/,PARM2/1.5/,PARM3/2.0/,PARM4/4.0/
      DATA DAMAX1/0.50/,DAMAX2/0.50/,NFPTS/16/
C    PARM1 IS USED TO COMPUTE RI.
C    THE PROGRAM ATTEMPTS TO CHOOSE A DT FOR A TIME STEP THAT WILL MAKE
C    THE POINT MOVE PARM2*RI.
C    WHEN INTERPOLATING SMOOTH CURVES FOR PLOTTING, DSMAX IS SET TO
C    PARM3*RI AND DAMAX IS SET TO DAMAX1.
C    WHEN INTERPOLATING NEW POINTS, DSMAX IS SET TO PARM4*RI AND DAMAX
```

```
C    IS SET TO DAMAX2.
C    NFPTS IS THE NUMBER OF POINTS INITIALLY USED TO DEFINE THE FRONT.
C    THE PROGRAM WILL ADD MORE POINTS AS THEY ARE NEEDED FOR GOOD
C    FRONT DEFINITION.  WE RECOMMENT NFPTS BE BETWEEN 12 AND 20.
      CALL PLOTS(0,0)
      CALL PLOT( 0.5, 2.0, -3 )
C
C    *****  DATA ITEM NUMBER 1, ITITLE  *****
C           FORMAT 20A4,/,20A4
C
C    ITITLE IS THE PLOT TITLE, ON TWO CARDS.  YOU MAY USE ALL EIGHTY
C    COLUMNS ON EACH OF THESE CARDS.
C
   10 READ(105,1000,END=990)(ITITLE(I),I=1,20)
      PRINT 1010,(ITITLE(I),I=1,20)
      READ  1000,(ITITLE(I),I=21,40)
      PRINT 1023,(ITITLE(I),I=21,40)
C
C    *****  DATA ITEM NUMBER 2, NWELLS, NTIME  *****
C           FORMAT 2I5
C
C    NWELLS IS THE TOTAL NUMBER OF WELLS, INCLUDING IMAGE WELLS IF ANY
C    ARE USED.  NWELLS MUST BE 1 THROUGH 99 INCLUSIVE.
C
C    NTIME IS THE NUMBER OF POINTS IN TIME (DATES) AT WHICH THE FRONT
C    LOCATIONS ARE TO BE PLOTTED.  NTIME MUST BE 0 THROUGH 20 INCLUSIVE.
C    IF NTIME IS ZERO, THE CALCULATION AND PLOTTING OF THE FRONTS
C    WILL BE SKIPPED.
C
      READ  1001,NWELLS,NTIME
      PRINT 1011,NWELLS,NTIME
C
C    *****  DATA ITEM NUMBER 3, QSLINE, H, PHI  *****
C           FORMAT 3F10.0
C
C    QSLINE DETERMINES HOW MANY STREAMLINES WILL EMANATE FROM EACH
C    INJECTION WELL.
C    IF QSLINE IS GREATER THAN ZERO, THE NUMBER OF STREAMLINES FOR A
```

```
C     INJECTION WELL IS CALCULATED BY DIVIDING THE WELL'S RATE IN
C     BARRELS PER DAY BY QSLINE AND ROUNDING THE RESULT TO THE NEAREST
C     INTEGER.  HOWEVER, NO INJECTION WELL WILL HAVE LESS THAN 10 OR
C     MORE THAN 100 STREAMLINES WHEN QSLINE IS POSITIVE.
C     IF QSLINE IS ZERO, THE CALCULATION AND PLOTTING OF STREAMLINES
C     WILL BE SKIPPED.
C     IF QSLINE IS LESS THAN ZERO, THE SAME NUMBER OF STREAMLINES WILL
C     EMANATE FROM EVERY INJECTION WELL.  THE NUMBER OF STREAMLINES PER
C     INJECTION WELL WILL EQUAL ABS(QSLINE).
C
C     H IS THE FORMATION THICKNESS IN FEET.
C
C     PHI IS THE FORMATION POROSITY EXPRESSED AS A FRACTION.
C
      READ  1002,QSLINE,H,PHI
      CONST=5.6146/(2.0*3.14159*H*PHI)
      PRINT 1015,QSLINE,H,PHI,CONST
      PRINT 1012
      U1=0.0
      DO 20 I=1,NWELLS
C
C     *****  DATA ITEM NUMBER 4, XW(I), YW(I), Q(I), I=1,NWELLS *****
C            NWELLS CARDS, ONE CARD FOR EACH WELL.  FORMAT 3F10.0
C
C     XW(I) IS THE X LOCATION OF WELL I.  ANY CARTESIAN COORDINATE
C     SYSTEM MAY BE USED TO SPECIFY THE WELL LOCATIONS.  THE UNITS
C     OF XW ARE FEET.
C
C     YW(I) IS THE Y LOCATION OF WELL I, IN FEET.
C
C     Q(I) IS THE RATE OF WELL I, IN BARRELS PER DAY.  POSITIVE Q IS
C     PRODUCTION, NEGATIVE Q IS INJECTION.
C
      READ 1002,XW(I),YW(I),Q(I)
      PRINT 1013,I,XW(I),YW(I),Q(I)
      BTTBAR(I)=0.0
   20 U1=U1+Q(I)
      PRINT 1014,U1
```

```
      IF(NTIME.LE.0)GO TO 22
      PRINT 1016
C
C    *****  DATA ITEM NUMBER 5, FPTIME(I), I=1,NTIME  *****
C           THIS ITEM WIL BE SKIPPED IF NTIME IS ZERO.  FORMAT 8F10.0
C
C    FPTIME SPECIFIES WHEN THE FRONT LOCATIONS ARE TO BE PLOTTED.
C    THE UNITS OF FPTIME ARE DAYS AFTER THE START OF INJECTION.
C    FPTIME(I+1) MUST BE GREATER THAN FPTIME(I) FOR ALL I.
C
      READ  1002,(FPTIME(ITIME),ITIME=1,NTIME)
      PRINT 1017,(FPTIME(ITIME),ITIME=1,NTIME)
      DO 21 ITIME=1,NTIME
      ATOTAL(ITIME)=0.0
      BITOT (ITIME)=0.0
   21 RBTOT (ITIME)=0.0
C
C    *****  DATA ITEM 6, XLEFT, XRIGHT, YLOWER, YUPPER, SCALE, RI  *****
C           FORMAT 6F10.0
C
C    XLEFT, XRIGHT, YLOWER, AND YUPPER SPECIFY THE REGION TO BE PLOTTED.
C    THEY HAVE UNITS OF FEET AND REFER TO THE COORDINATE SYSTEM USED TO
C    SPECIFY WELL LOCATIONS.  THE LEFT EDGE OF THE PLOT IS LOCATED AT
C    X = XLEFT AND THE RIGHT EDGE IS AT X = XRIGHT.  THE BOTTOM EDGE OF
C    THE PLOT IS AT Y = YLOWER AND THE TOP EDGE IS AT Y = YUPPER.  XLEFT
C    MUST NOT EQUAL XRIGHT.  XLEFT GREATER THAN XRIGHT IS O.K.  YLOWER
C    MUST NOT EQUAL YUPPER.  YLOWER GREATER THAN YUPPER IS O.K.  ONLY
C    THOSE PORTIONS OF FRONTS AND STREAMLINES THAT ARE INSIDE THE PLOT
C    REGION ARE PLOTTED.  HOWEVER, ALL OF THE WELLS ARE USED IN THE
C    CALCULATION OF THE FRONTS AND STREAMLINES.  THE FRONTS AND STREAM-
C    LINES THAT EMANATE FROM WELLS THAT ARE OUTSIDE OF THE PLOT REGION
C    ARE NOT PLOTTED EVEN WHEN PORTIONS OF THESE FRONTS AND STREAMLINES
C    LIE INSIDE THE PLOT REGION.
C
C    SCALE IS THE PLOT SCALE IN FEET OF REGION PER INCH OF PLOT.  THE
C    PLOT WILL BE ABS(XRIGHT-XLEFT)/SCALE INCHES WIDE.  IT WILL BE
C    ABS(YUPPER-YLOWER)SCALE INCHES HIGH.
C
```

```
C    RI IS THE DISTANCE USED BY THE PROGRAM FOR A SMALL DISPLACEMENT
C    IN THE RESERVOIR.  RI IS IN FEET.  IF RI IS INPUT AS ZERO, THE
C    PROGRAM WILL CALCULATE THE DISTANCE BETWEEN THE CLOSEST PAIR OF
C    WELLS.  RI IS THEN SET TO 0.025 OF THIS DISTANCE.  THIS DEFAULT
C    VALUE SEEMS TO WORK WELL.
C
   22 READ  1002,XLEFT,XRIGHT,YLOWER,YUPPER,SCALE,RI
      PRINT 1018,XLEFT,XRIGHT,YLOWER,YUPPER,SCALE,RI
      XMIN=AMIN1(XLEFT,XRIGHT)
      XMAX=AMAX1(XLEFT,XRIGHT)
      YMIN=AMIN1(YLOWER,YUPPER)
      YMAX=AMAX1(YLOWER,YUPPER)
      SCALE=AMAX1(((XMAX-XMIN)/52.0),((YMAX-YMIN)/26.0),SCALE)
      WPLOT=(XMAX-XMIN)/SCALE
      HPLOT=(YMAX-YMIN)/SCALE
      IF(RI.GT.0.0)GO TO 50
      U1=(XMAX-XMIN)**2+(YMAX-YMIN)**2
      DO 40 I=1,NWELLS
      XBAR=XW(I)
      IF((XBAR.LT.XMIN).OR.(XBAR.GT.XMAX))GO TO 40
      YBAR=YW(I)
      IF((YBAR.LT.YMIN).OR.(YBAR.GT.YMAX))GO TO 40
      DO 30 J=1,NWELLS
      IF(J.EQ.I)GO TO 30
      U1=AMIN1(((XW(J)-XBAR)**2+(YW(J)-YBAR)**2),U1)
   30 CONTINUE
   40 CONTINUE
      RI=PARM1*SQRT(U1)
      PRINT 1024,RI
   50 IF(RI.LE.0.05*SCALE)GO TO 990
      RISQ=RI*RI
      XPPARM(1)=XLEFT
      XPPARM(2)=SCALE
      IF(XRIGHT.LT.XLEFT)XPPARM(2)=-SCALE
      XPPARM(3)=XMIN
      XPPARM(4)=XMAX
      YPPARM(1)=YLOWER
      YPPARM(2)=SCALE
```

```
      IF(YUPPER.LT.YLOWER)YPPARM(2)=-SCALE
      YPPARM(3)=YMIN
      YPPARM(4)=YMAX
      DSMAX1=PARM3*RI
      DSMAX2=PARM4*RI
      CALL PLOT(WPLOT, 0.0, 2 )
      CALL PLOT(WPLOT,HPLOT, 2 )
      CALL PLOT( 0.0, HPLOT, 2 )
      CALL PLOT( 0.0, 0.0, 2 )
      U1=HPLOT+0.42
      CALL SYMBOL( 0.0, U1, 0.14, ITITLE, 0.0, 80 )
      U1=U1-0.28
      CALL SYMBOL( 0.0, U1, 0.14, ITITLE(21), 0.0, 80 )
      DO 400 I=1,NWELLS
      XBAR=XW(I)
      IF((XBAR.LT.XMIN).OR.(XBAR.GT.XMAX))GO TO 400
      YBAR=YW(I)
      IF((YBAR.LT.YMIN).OR.(YBAR.GT.YMAX))GO TO 400
      U1=(XBAR-XMIN)/SCALE
      U2=(YBAR-YMIN)/SCALE
      CALL SYMBOL(U1,U2, 0.07, 1, 0.0, -1)
      IF(Q(I).GT.0.0)CALL SYMBOL(U1,U2, 0.07, 11, 0.0, -1)
      IF((NTIME.EQ.0).AND.(QSLINE.EQ.0.0))CALL NUMBER(U1+0.1,U2+0.1,
     2 0.14,Q(I),0.0,2)
      IF(Q(I).GE.0.0)GO TO 400
      U1=U1+0.07
      U2=U2+0.07
      CALL PLOT(U1,U2,3)
      U1=U1-0.14
      U2=U2-0.14
      CALL PLOT(U1,U2, 2 )
      TZERO=-RISG/(2.0*Q(I)*CONST)
      IF(NTIME.LE.0)GO TO 300
      DO 100 J=1,NWELLS
  100 BTTIME(J)=0.0
      DTHETA=6.283185/FLOAT(NFPTS)
      THETA=0.5*DTHETA
      DO 110 K=1,NFPTS
```

```
      KPOLD(K)=0
      XFOLD(K)=XBAR+RI*COS(THETA)
      YFOLD(K)=YBAR+RI*SIN(THETA)
      CALL PTRACK(0.0,XFOLD(K),YFOLD(K),DXDTO(K),DYDTO(K),KPOLD(K))
  110 THETA=THETA+DTHETA
      PRINT 1020,I
      NFPOLD=NFPTS
      DO 290 ITIME=1,NTIME
      NN=0
      DO 190 K=1,NFFOLD
      KPROD=KPOLD(K)
      XBAR=XFOLD(K)
      YBAR=YFOLD(K)
      DXDT=DXDTO(K)
      DYDT=DYDTO(K)
      IF(KPROD.LT.0)GO TO 160
      TIME=TZERO
      IF(ITIME.GT.1)TIME=FPTIME(ITIME-1)
  120 DT=PARM2*RI/SQRT(DXDT**2+DYDT**2)
      U1=(FPTIME(ITIME)-TIME)/DT
      IF(U1.GE.1.6)GO TO 140
      IF(U1.GT.1.1)GO TO 130
      DT=FPTIME(ITIME)-TIME
      GO TO 140
  130 DT=0.5*(FPTIME(ITIME)-TIME)
  140 U1=XBAR
      U2=YBAR
      CALL PTRACK(DT,XBAR,YBAR,DXDT,DYDT,KPROD)
      TIME=TIME+DT
      IF(KPROD.LT.0)GO TO 150
      IF((FPTIME(ITIME)-TIME).GT.(0.01*DT))GO TO 120
      JBAR=1
      GO TO 170
  150 KBAR=-KPROD
      IF(BTTIME(KBAR).NE.0.0)GO TO 160
      BTTIME(KBAR)=TIME
      PRINT 1021,KBAR,TIME
  160 IF((NN.GT.0).AND.(KPNEW(NN-1).EQ.KPROD))GO TO 190
```

```
      JBAR=2
  170 CONTINUE
      DO 180 J=1,JBAR
      IF(NN.LT.121)GO TO 172
      PRINT 1019,NN,I,FPTIME(ITIME),K
      CALL EXIT
  172 NN=NN+1
      XFNEW(NN)=XBAR
      YFNEW(NN)=YBAR
      DXDTN(NN)=DXDT
      DYDTN(NN)=DYDT
  180 KPNEW(NN)=KPRED
  190 CONTINUE
      IF(KPNEW(1).GE.0)GO TO 210
  200 IF(KPNEW(NN).NE.KPNEW(1))GO TO 210
      NN=NN-1
      IF(NN.GT.2)GO TO 200
  210 IF(NN.LE.2)GO TO 300
      NFPNEW=NN+1
      XFNEW(NFPNEW)=XFNEW(1)
      YFNEW(NFPNEW)=YFNEW(1)
      KBAR=NN
      DO 220 K=1,NN
C     IF K IS EQUAL TO 1, KBAR=NN.  IF K IS GREATER THAN 1, KBAR=K-1.
      ANGLE(K)=ATAN2((YFNEW(K+1)-YFNEW(KBAR)),(XFNEW(K+1)-XFNEW(KBAR)))
  220 KBAR=K
      ANGLE(NFPNEW)=ANGLE(1)
      CALL CURVIT(NFPNEW,XFNEW,YFNEW,ANGLE,DSMAX1,DAMAX1,500,NCURVE,
     2 XCURVE,YCURVE)
      DO 230 K=1,4
      XCURVE(NCURVE+K)=XPPARM(K)
  230 YCURVE(NCURVE+K)=YPPARM(K)
      CALL LINEZ1(XCURVE,YCURVE,NCURVE)
      DSUM1=0.0
      KBAR=NCURVE-1
      DO 240 K=1,KBAR
  240 DSUM1=(DSUM1+XCURVE(K)*YCURVE(K+1))-XCURVE(K+1)*YCURVE(K)
      AREA=0.5*DSUM1/43560.0
```

```
      RESBBL=0.5*DSUM1*H*PHI/5.6146
      BBLINJ=-FPTIME(ITIME)*Q(1)
      PRINT 1022,FPTIME(ITIME),BBLINJ,I,AREA,RESBBL
      ATOTAL(ITIME)=ATOTAL(ITIME)+AREA
      RBTOT(ITIME)=RBTOT(ITIME)+RESBBL
      BITOT(ITIME)=BITOT(ITIME)+BBLINJ
      CALL CURVIT(NFPNEW,XFNEW,YFNEW,ANGLE,DSMAX2,DAMAX2,121,NFPOLD,
     2 XFOLD,YFOLD)
      NFPOLD=NFPOLD-1
      NN=1
      DO 260 K=1,NFPOLD
      IF((XFOLD(K).EQ.XFNEW(NN)).AND.(YFOLD(K).EQ.YFNEW(NN)))GO TO 250
      CALL PTRACK(0.0,XFOLD(K),YFOLD(K),DXDTO(K),DYDTO(K),KPOLD(K))
      GO TO 260
  250 DXDTO(K)=DXDTN(NN)
      DYDTO(K)=DYDTN(NN)
      KPOLD(K)=KPNEW(NN)
      NN=NN+1
      IF(NN.GT.NFPNEW)NN=1
  260 CONTINUE
  290 CONTINUE
      DO 292 J=1,NWELLS
      IF(BTTIME(J).EQ.0.0) GO TO 292
      IF(BTTBAR(J).GT.0.0)BTTBAR(J)=AMIN1(BTTBAR(J),BTTIME(J))
      IF(BTTBAR(J).EQ.0.0)BTTBAR(J)=BTTIME(J)
  292 CONTINUE
  300 IF(QSLINE)320,400,310
  310 NSLINE=IFIX(0.5-Q(I)/QSLINE)
      IF(NSLINE.LT.10)NSLINE=10
      GO TO 330
  320 NSLINE=IFIX(0.01-QSLINE)
  330 IF(NSLINE.GT.100)NSLINE=100
      DTHETA=6.283185/FLOAT(NSLINE)
      THETA=0.5*DTHETA
      DO 390 ISLINE=1,NSLINE
      XBAR=XW(I)+RI*COS(THETA)
      YBAR=YW(I)+RI*SIN(THETA)
      KPROD=0
```

```
      CALL PTRACK(0.0,XBAR,YBAR,DXDT,DYDT,KPROD)
      NN=0
  340 NN=NN+1
      XFNEW(NN)=XBAR
      YFNEW(NN)=YBAR
      ANGLE(NN)=ATAN2(DYDT,DXDT)
      IF((XBAR.LT.XMIN).OR.(XBAR.GT.XMAX))GO TO 370
      IF((YBAR.LT.YMIN).OR.(YBAR.GT.YMAX).OR.(NN.GE.361))GO TO 370
      DT=PARM2*RI/SQRT(DXDT**2+DYDT**2)
      CALL PTRACK(DT,XBAR,YBAR,DXDT,DYDT,KPROD)
      IF(KPROD.GE.0)GO TO 340
  350 U1=(XFNEW(NN)-XBAR)**2+(YFNEW(NN)-YBAR)**2
      IF(U1.GT.RISQ)GO TO 360
      NN=NN-1
      GO TO 350
  360 U1=RI/SQRT(U1)
      NN=NN+1
      XFNEW(NN)=XBAR+U1*(XFNEW(NN-1)-XBAR)
      YFNEW(NN)=YBAR+U1*(YFNEW(NN-1)-YBAR)
      CALL PTRACK(0.0,XFNEW(NN),YFNEW(NN),DXDT,DYDT,KPROD)
      ANGLE(NN)=ATAN2(DYDT,DXDT)
  370 CONTINUE
      NFPNEW=NN
      CALL CURVIT(NFPNEW,XFNEW,YFNEW,ANGLE,RI,PARM5,500,NCURVE,XCURVE,
     2 YCURVE)
      DO 380 K=1,4
      XCURVE(NCURVE+K)=XPPARM(K)
  380 YCURVE(NCURVE+K)=YPPARM(K)
      CALL LINEZ1(XCURVE,YCURVE,NCURVE)
  390 THETA=THETA+DTHETA
  400 CONTINUE
      PRINT 1041
      PRINT 1040,(FPTIME(I),ATOTAL(I),RBTOT(I),BITOT(I),I=1,NTIME)
      DO 500 J=1,NWELLS
      IF(BTTBAR(J).EQ.0.0)GO TO 500
      PRINT 1021,J,BTTBAR(J)
  500 CONTINUE
      U1=WPLOT+4.0
```

```
      CALL PLOT(U1, 0.0, -3 )
      GO TO 10
  990 CALL PLOT( -3.5, -2.0, 999 )
      PRINT 1030
      CALL EXIT
 1000 FORMAT(20A4)
 1001 FORMAT(16I5)
 1002 FORMAT(8F10.0)
 1010 FORMAT(1H1,9X,20A4)
 1011 FORMAT(1H0,9X,8HNWELLS =,I5,/,10X,7HNTIME =,I5)
 1012 FORMAT(48H0          WELL       X LOC        Y LOC        RATE,/,
     2 48H          NO        FEET          FEET          RBPD)
 1013 FORMAT(I13,2F12.1,F12.2)
 1014 FORMAT(19H0          Q TOTAL =,F12.2)
 1015 FORMAT(1H0,9X,8HQSLINE =,F12.5,9H BBLS/DAY,/,10X,8HH       =,
     2 F12.5,5H FEET,/,10X,8HPHI     =,F12.5,11H (FRACTION),/,10X,
     3 8HCONST  =,F12.5)
 1016 FORMAT(55H0          FPTIME VECTOR (TIMES FOR PLOTTING FRONTS, IN ,
     2 5HDAYS))
 1017 FORMAT(5X,8F10.2)
 1018 FORMAT(1H0,9X,7HXLEFT =,F7.0,5X,8HXRIGHT =,F7.0,5X,8HYLOWER =,
     2 F7.0,5X,8HYUPPER =,F7.0,5X,7HSCALE =,F7.2,5X,4HRI =,F7.2)
 1019 FORMAT(53H0**** FATAL ERROR, THE NUMBER OF POINTS ON THE FRONT ,
     2 46HEXCEECED AVAILABLE STORAGE WHILE COMPUTING THE,/,
     3 53H *****  LOCATION OF THE FRONT SURROUNDING WELL NUMBER,I4,
     4 11H FOR TIME =,F8.2,12H DAYS.   (K =,I4,6H, NN =,I4,1H))
 1020 FORMAT(51H1          CALCULATION OF THE FRONT SURROUNDING THE ,
     2 32HFLUIDS INJECTED INTO WELL NUMBER,I4)
 1021 FORMAT(50H0          THE FRONT BROKE THROUGH INTO WELL NUMBER,I4,
     2 3H AT,F8.2,6H DAYS.)
 1022 FORMAT(15H0          AFTER,F8.2,17H DAYS, A TOTAL OF,F10.1,
     2 44H BARRELS HAVE BEEN INJECTED INTO WELL NUMBER,I4,1H.,/,
     3 10X,32HAT THIS TIME, THE FRONT INCLOSES,F10.3,13H ACRES WHICH ,
     4 7HCONTAIN,F10.1,19H RESERVOIR BARRELS.)
 1023 FORMAT(10X,20A4)
 1024 FORMAT(1H0,9X,21HDEFAULT VALUE OF RI =,F12.5,5H FEET)
 1030 FORMAT(20H0          END OF RUN)
 1040 FORMAT(T9,F8.2,T25,F10.1,T44,F10.1,T63,F10.1/)
```

```
 1041 FORMAT(1H1,T10,'TIME',T25,'TOTAL AREA',T40,'TOTAL RESERVOIR',T60,
     2 'TOTAL INJECTED',/,T10,'DAYS',T28,'ACRES',T45,'BARRELS',T64,'BARR
     2ELS'/)
      END
C
      SUBROUTINE CURVIT(NN,X,Y,ANGLE,DSMAX,DAMAX,NIMAX,NINT,XINT,YINT)
C
C    SUBROUTINE CURVIT INTERPOLATES A SMOOTH CURVE THROUGH A SERIES OF
C    POINTS.  CURVIT REQUIRES THE TANGENT DIRECTIONS AT THE DATA POINTS.
C
C    INPUT PARAMETERS:
C
C    NN = NUMBER OF DATA POINTS.
C
C    X IS THE ARRAY CONTAINING THE ABCISSAS OF THE DATA POINTS.
C
C    Y IS THE ARRAY CONTAINING THE ORDINATES OF THE DATA POINTS.
C
C    ANGLE IS THE ARRAY CONTAINING THE TANGENT DIRECTIONS OF THE CURVE
C    (IN RADIANS) AT THE DATA POINTS.  **** WARNING, ANGLE(I) MUST
C    POINT IN THE DIRECTION OF THE CURVE SEGMENT GOING TOWARD THE I+1
C    DATA POINT.  AN ERROR OF PI RADIANS (180 DEGREES) FOR ANGLE(I)
C    WILL MAKE IT POINT TOWARD THE I-1 DATA POINT. *****  THE ANGLE
C    OF THE CHORD BETWEEN THE I-1 AND THE I+1 DATA POINTS IS OFTEN A
C    GOOD APPROXIMATION TO THE TANGENT DIRECTION AT POINT I.  I.E.
C    ANGLE(I)=ATAN((Y(I+1)-Y(I-1)),(X(I+1),X(I-1)))
C
C    DSMAX IS THE MAXIMUM PERMISSIBLE SEPARATION BETWEEN TWO ADJACENT
C    INTERPOLATED POINTS.  DSMAX HAS THE SAME UNITS AS X AND Y.
C
C    DAMAX IS THE MAXIMUM PERMISSIBLE CHANGE IN TANGENT DIRECTION
C    BETWEEN TWO ADJACENT INTERPOLATED POINTS.  DSMAX IS IN RADIANS.
C    THIS DAMAX CRITERION MAY NOT BE MEET IF TWO ADJACENT DATA POINTS
C    ARE CLOSER THAN DSMAX AND THE CHANGE IN TANGENT DIRECTION BETWEEN
C    THESE POINTS IS LARGER THAN DAMAX.
C
C    NIMAX IS THE MAXIMUM ALLOWABLE NUMBER OF POINTS ON THE SMOOTH
C    CURVE.  NIMAX SHOULD BE LESS THAN OR EQUAL TO THE DIMENSION OF THE
```

```
C     XINT AND YINT ARRAYS.  THIS WILL PREVENT OVERSTORING IF THE INPUT
C     VALUES OF DSMAX AND DAMAX RESULT IN TOO MANY INTERPOLATED POINTS.
C
C     OUTPUT PARAMETERS:
C
C     NINT = NUMBER OF POINTS IN THE OUTPUT SMOOTH CURVE.  ALL INPUT DATA
C     POINTS WILL BE ON THIS CURVE.  IN ADDITION, CURVIT WILL INTERPOL-
C     ATE BETWEEN INPUT DATA POINTS WHENEVER NECESSARY TO MEET THE DSMAX
C     AND DAMAX REQUIREMENTS.  IF NIMAX IS EXCEEDED, AN ERROR MESSAGE
C     WILL BE PRINTED AND NINT WILL BE SET TO NIMAX + 1.  HOWEVER, ONLY
C     NIMAX POINTS WILL BE RETURNED IN XINT AND YINT.
C
C     XINT IS AN ARRAY OF ABCISSAS OF POINTS ON THE SMOOTH CURVE.
C
C     YINT IS AN ARRAY OF ORDINATES OF POINTS ON THE SMOOTH CURVE.
C
C     BETWEEN EACH ADJACENT PAIR OF DATA POINTS, CURVIT APPROXIMATES X AS
C     AN HERMITE CUBIC IN ARC LENGTH S.  IT APPROXIMATES Y AS A SECOND
C     INDEPENDENT HERMITE CUBIC IN S.  IT APPROXIMATES ARC LENGTH AS THE
C     CHORD LENGTH BETWEEN THE DATA POINTS.  COMMENT CARDS INDICATE THE
C     CHANGES THAT ARE REQUIRED TO OBTAIN A MORE ACCURATE APPROXIMATION
C     FOR ARC LENGTH.  THESE REFINMENTS GIVE SLIGHTLY BETTER CURVES AT
C     THE EXPENSE OF CONSIDERABLE ADDITIONAL COMPUTATION.
C
C     WRITTEN BY C. F. ZETIK, CITIES SERVICE OIL COMPANY, APRIL 1975.
C
      DIMENSION X(1),Y(1),ANGLE(1),XINT(1),YINT(1)
      IF(NN.GT.0)GO TO 10
         NINT=0
         RETURN
   10 K=0
      JMAX=0
      DO 40 I=1,NN
         XB=X(I)
         YB=Y(I)
         DXDSB=COS(ANGLE(I))
         DYDSB=SIN(ANGLE(I))
         IF(I.EQ.1)GO TO 30
```

```
              ILESS1=I-1
              DELTAS=SQRT((XB-XA)**2+(YB-YA)**2)
              DANGLE=ABS(ANGLE(I)-ANGLE(ILESS1))
              IF(DANGLE.GT.3.141593)DANGLE=6.283185-DANGLE
              U=DELTAS/DSMAX
              IF(U.LT.1.0)JMAX=IFIX(U*(1.0+DANGLE/DAMAX))
              IF(U.GE.1.0)JMAX=IFIX(U+DANGLE/DAMAX)
              IF(JMAX.EQ.0)GO TO 30
   15            CX1=DELTAS*DXDSA
                 CY1=DELTAS*DYDSA
                 CX2=3.0*(XB-XA)-DELTAS*(2.0*DXDSA+DXDSB)
                 CY2=3.0*(YB-YA)-DELTAS*(2.0*DYDSA+DYDSB)
                 CX3=-2.0*(XB-XA)+DELTAS*(DXDSA+DXDSB)
                 CY3=-2.0*(YB-YA)+DELTAS*(DYDSA+DYDSB)
                 DELTAU=1.0/FLOAT(JMAX+1)
                 U=DELTAU
C                DSOLD=DELTAS
C                DELTAS=0.0
                 L=K
                 DO 20 J=1,JMAX
                    L=L+1
                    IF(L.GT.NIMAX)GO TO 50
                    XINT(L)=XA+U*(CX1+U*(CX2+U*CX3))
                    YINT(L)=YA+U*(CY1+U*(CY2+U*CY3))
C                   DELTAS=DELTAS+SQRT((XINT(L)-XINT(L-1))**2
C     2              +(YINT(L)-YINT(L-1))**2
                    U=U+DELTAU
   20         CONTINUE
C                DELTAS=DELTAS+SQRT((XB-XINT(L))**2+(YB-YINT(L))**2)
C                IF(ABS(1.0-DELTAS/DSOLD).GT.0.01)GO TO 15
   30       K=K+JMAX+1
            IF(K.GT.NIMAX)GO TO 60
            XINT(K)=XB
            YINT(K)=YB
            XA=XB
            YA=YB
            DXDSA=DXDSB
            DYDSA=DYDSB
```

```
   40 CONTINUE
      NINT=K
      RETURN
   50 K=L
   60 NINT=K
      I=ILESS1+1
      PRINT 1000,NIMAX,ILESS1,I
      RETURN
 1000 FORMAT(43H0*****  ERROR IN SUBROUTINE CURVIT, NIMAX =,I12,4H WAS,
     2 9H EXCEEDED,//,42H *****  WHILE INTERPOLATING BETWEEN POINTS,I12,
     3 4H AND,I12,/)
      END
C
      SUBROUTINE LINEZ1(XARRAY,YARRAY,NPTS)
C
C   SUBROUTINE LINEZ1 PLOTS A SERIES OF POINTS CONNECTED BY LINE SEGE-
C   MENTS.  IT IS SIMILAR TO CALCOMPS SUBROUTINE LINE EXCEPT IT CHECKS
C   AGAINST LIMITS AND ONLY PLOTS POINTS AND THE PARTS OF LINE SEGEMENTS
C   THAT ARE WITHIN THE LIMITS.
C
C   WRITTEN BY D. F. ZETIK, CITIES SERVICE OIL COMPANY, APRIL, 1975.
C
      DIMENSION XARRAY(1),YARRAY(1)
      XFIRST=XARRAY(NPTS+1)
      DELTAX=XARRAY(NPTS+2)
      U1    =(XARRAY(NPTS+3)-XFIRST)/DELTAX
      U2    =(XARRAY(NPTS+4)-XFIRST)/DELTAX
      XPLEFT=AMIN1(U1,U2)
      XPRGHT=AMAX1(U1,U2)
      YFIRST=YARRAY(NPTS+1)
      DELTAY=YARRAY(NPTS+2)
      U1    =(YARRAY(NPTS+3)-YFIRST)/DELTAY
      U2    =(YARRAY(NPTS+4)-YFIRST)/DELTAY
      YPBTTM=AMIN1(U1,U2)
      YPTOP=AMAX1(U1,U2)
C   FIND (XPLOT,YPLOT), THE CURRENT LOCATION OF THE PEN.
      CALL WHERE(XPLOT,YPLOT,DUMMY)
C   DETERMINE WHICH END OF THE LINE IS CLOSER TO (XPLOT,YPLOT).
```

```
      U1=(XPLOT-(XARRAY(1)-XFIRST)/DELTAX)**2+
     2 (YPLOT-(YARRAY(1)-YFIRST)/DELTAY)**2
      U2=(XPLOT-(XARRAY(NPTS)-XFIRST)/DELTAX)**2+
     2 (YPLOT-(YARRAY(NPTS)-YFIRST)/DELTAY)**2
      K=1
      KSTEP=1
      IF(U2.LT.U1)K=NPTS
      IF(U2.LT.U1)KSTEP=-1
C    IF IPEN=2, THE PLOTTER PEN IS DOWN.  IF IPEN=3, THE PEN IS UP.
      IPEN=3
      DO 90 I=1,NPTS
         XPLOT=(XARRAY(K)-XFIRST)/DELTAX
         YPLOT=(YARRAY(K)-YFIRST)/DELTAY
         IF((XPLOT.LT.XPLEFT).OR.(XPLOT.GT.XPRGHT))GO TO 10
         IF((YPLOT.LT.YPBTTM).OR.(YPLOT.GT.YPTOP))GO TO 10
         IF((I.EQ.1).OR.(IPEN.EQ.2))GO TO 70
C    POINT K IS INSIDE THE PLOT AREA, POINT KLAST IS OUTSIDE.
         XBAR=(XARRAY(KLAST)-XFIRST)/DELTAX
         YBAR=(YARRAY(KLAST)-YFIRST)/DELTAY
         GO TO 20
   10    IF(IPEN.EQ.3)GO TO 80
C    POINT K IS OUTSIDE THE PLOT AREA, POINT KLAST IS INSIDE.
         XBAR=XPLOT
         YBAR=YPLOT
         XPLOT=(XARRAY(KLAST)-XFIRST)/DELTAX
         YPLOT=(YARRAY(KLAST)-YFIRST)/DELTAY
C    CONNECT A STRAIGHT LINE BETWEEN THE POINT (XBAR,YBAR), WHICH IS
C    OUTSIDE THE PLOT AREA, AND THE POINT (XPLOT,YPLOT), WHICH IS INSIDE
C    THE PLOT AREA.  REPLACE (XBAR,YBAR) WITH THE POINT WHERE THIS
C    STRAIGHT LINE INTERSECTS THE BORDER OF THE PLOT AREA.
   20    IF(XBAR.GE.XPLEFT)GO TO 30
         YBAR=(YBAR*(XPLOT-XPLEFT)+YPLOT*(XPLEFT-XBAR))/(XPLOT-XBAR)
         XBAR=XPLEFT
         GO TO 40
   30    IF(XBAR.LE.XPRGHT)GO TO 40
         YBAR=(YBAR*(XPRGHT-XPLOT)+YPLOT*(XBAR-XPRGHT))/(XBAR-XPLOT)
         XBAR=XPRGHT
   40    IF(YBAR.GE.YPBTTM)GO TO 50
```

```
          XBAR=(XBAR*(YPLOT-YPBTTM)+XPLOT*(YPBTTM-YBAR))/(YPLOT-YBAR)
          YBAR=YPBTTM
          GO TO 60
   50     IF(YBAR.LE.YPTOP)GO TO 60
          XBAR=(XBAR*(YPTOP-YPLOT)+XPLOT*(YBAR-YPTOP))/(YBAR-YPLOT)
          YBAR=YPTOP
   60     CALL PLOT(XBAR,YBAR,IPEN)
          IPEN=5-IPEN
          IF(IPEN.EQ.3)GO TO 80
   70     CALL PLOT(XPLOT,YPLOT,IPEN)
          IPEN=2
   80     KLAST=K
          K=K+KSTEP
   90 CONTINUE
      RETURN
      END
C
      SUBROUTINE PTRACK(DT,XVAL,YVAL,DXDT,DYDT,KPROD)
C
C   SUBROUTINE PTRACK TRACKS A POINT IN THE FLUID WITHIN A HOMOGENEOUS,
C   ISOTROPIC, CONSTANT THICKNESS RESERVOIR.
C
C   WRITTEN BY D. F. ZETIK, CITIES SERVICE OIL COMPANY, APRIL, 1975.
C
      COMMON/WDATA/Q(99),XW(99),YW(99),BTTIME(99),NWELLS,CONST,RI,RISQ
      DIMENSION XK(4),YK(4),BB(4)
      DATA BB/0.1666667,0.3333333,0.3333333,0.1666667/
      DOUBLE PRECISION DSUM1,DSUM2
C
C   *****  PART 1, TIME STEP MODIFICATION  *****
C
      IF(DT.EQ.0.0)GO TO 90
         IF((KPROD.LE.0).OR.(Q(KPROD).LE.0.0))GO TO 30
C           ASSUME RADIAL FLOW INTO WELL KPROD AND NEGLECT THE EFFECTS
C           OF ALL OTHER WELLS.  CALCULATE TTPROD, THE TIME REQUIRED
C           FOR THE POINT TO REACH WELL KPROD.  U1 IS THE SQUARE OF THE
C           DISTANCE TO WELL KPROD.  RISQ IS THE SQUARE OF RI.
            U1=(XW(KPROD)-XVAL)**2+(YW(KPROD)-YVAL)**2
```

```
            TTPROD=U1/(2.0*Q(KPROD)*CONST).
            IF(U1.GT.RISQ)GO TO 20
               IF(DT.LT.TTPROD)GO TO 10
C                 THE DISTANCE FROM THE POINT TO WELL KPROD IS LESS
C                 THAN RI AND TTPROD IS LESS THAN DT.  SET DT = TTPROD
C                 AND PRODUCE THE POINT FROM WELL KPROD
                  DT=TTPROD
                  XVAL=XW(KPROD)
                  YVAL=YW(KPROD)
                  DXDT=0.0
                  DYDT=0.0
                  KPROD=-KPROD
               RETURN
C                 THE DISTANCE FROM THE POINT TO WELL KPROD IS LESS
C                 THAN RI BUT TTPROD IS GREATER THAN DT.  ASSUME RADIAL
C                 FLOW INTO WELL KPROD AND NEGLECT THE EFFECTS OF ALL
C                 OTHER WELLS TO CALCULATE THE POSITION OF THE POINT
C                 AT THE END OF THE TIME STEP.
   10             U1=(1.0-DT/TTPROD)**2
                  XVAL=XW(KPROD)+U1*(XVAL-XW(KPROD))
                  YVAL=YW(KPROD)+U1*(YVAL-YW(KPROD))
               GO TO 90
C           THE DISTANCE FROM THE POINT TO WELL KPROD IS GREATER THAN
C           RI.  IF DT IS GREATER THAN 0.5*TTPROD, REDUCE DT.  THE
C           CALLING PROGRAM MAY REQUIRE POINT LOCATIONS AT SPECIFIED
C           TIMES.  THUS, THE NEXT VALUE OF DT MAY EQUAL THE PRESENT
C           REDUCTION IN DT.  BY MAKING THE REDUCTION IN DT AT LEAST
C           33.3 PERCENT, WE ELIMINATE A POSSIBLE CAUSE OF VERY SMALL
C           TIME STEPS.
   20       IF(TTPROD.LT.(2.0*DT))DT=AMIN((0.5*TTPROD),(0.667*DT))
C
C  *****  PART 2, RUNGE-KUTTA INTEGRATION OF POINT'S TRAJECTORY  *****
C
   30    XK(1)=DXDT*DT
         YK(1)=DYDT*DT
         DO 70 ISTEP=2,4
            IF(ISTEP.EQ.4)GO TO 40
               XBAR=XVAL+0.5*XK(ISTEP-1)
```

```
            YBAR=YVAL+0.5*YK(ISTEP-1)
         GO TO 50
   40       XBAR=XVAL+XK(ISTEP-1)
            YBAR=YVAL+YK(ISTEP-1)
   50    DSUM1=0.0
         DSUM2=0.0
         DO 60 I=1,NWELLS
            U1=Q(I)/((XW(I)-XBAR)**2+(YW(I)-YBAR)**2)
            DSUM1=DSUM1+(XW(I)-XBAR)*U1
            DSUM2=DSUM2+(YW(I)-YBAR)*U1
   60    CONTINUE
         XK(ISTEP)=CONST*DSUM1*DT
         YK(ISTEP)=CONST*DSUM2*DT
   70 CONTINUE
C     REPLACE INPUT POINT LOCATION WITH THE COMPUTED LOCATION
C     AFTER TIME INTERVAL DT.
      DO 80 ISTEP=1,4
         XVAL=XVAL+BB(ISTEP)*XK(ISTEP)
         YVAL=YVAL+BB(ISTEP)*YK(ISTEP)
   80 CONTINUE
C
C  *****  PART 3, COMPUTE DXDT AND DYDT, THE POINT'S VELOCITY  *****
C         VECTOR COMPONENTS.  ALSO COMPUTE KPROD, THE INDEX OF
C         THE PRODUCTION WELL THAT EXERTS THE GREATEST INFLUENCE
C         ON THE POINT.
C
   90 KPROD=0
      TEST=0.0
      DSUM1=0.0
      DSUM2=0.0
      DO 140 I=1,NWELLS
         U1=Q(I)/((XW(I)-XVAL)**2+(YW(I)-YVAL)**2)
         IF(U1.GT.TEST)KPROD=I
         IF(U1.GT.TEST)TEST=U1
         DSUM1=DSUM1+(XW(I)-XVAL)*U1
         DSUM2=DSUM2+(YW(I)-YVAL)*U1
  140 CONTINUE
      DXDT=CONST*DSUM1
```

```
      DYDT=CONST*DSUM2
      RETURN
      END
*STOP* 0
```